Integration der Normalparabel

Prof. Dr. Dörte Haftendorn, MuPAD 4, http://haftendorn.uni-lueneburg.de Aug.06

Automatische Übersetzung aus MuPAD 3.11,Juni 06 Update 21.06.06

Es fehlen nocht textliche Änderungen, die MuPAD 4 direkt berücksichtigen, das ist in Arbeit.

Web: http://haftendorn.uni-lueneburg.de www.mathematik-verstehen.de

++

```
f:=x->x^2
```

$x \to x^2$

```
summen:=student::plotRiemann(f(x),x=0..4,10):plot(summen)
```




```
summen:=student::plotRiemann(f(x),x=0..4,20):
plot(summen)
```




```
summen:=student::plotRiemann(f(x),x=0..4,50):
plot(summen)
```



Bildung der Unter- und Obersumme

```
sum(i^2, i=1..n)
```

$$\frac{n\cdot(n+1)\cdot(2\cdot n+1)}{6}$$

```
sum(i^2, i=1..n-1)
```

$$\frac{n\cdot(n-1)\cdot(2\cdot n-1)}{6}$$

```
untersumme:=b^3/n^3*sum(i^2, i=0..n-1)
```

$$\frac{b^3\cdot(n-1)\cdot(2\cdot n-1)}{6\cdot n^2}$$

```
obersumme:=b^3/n^3*sum(i^2, i=1..n)
```

$$\frac{b^3\cdot(n+1)\cdot(2\cdot n+1)}{6\cdot n^2}$$

```
limit(expand(untersumme), n=infinity)
```

$$\frac{b^3}{3}$$

```
limit(expand(obersumme), n=infinity)
```

$$\frac{b^3}{3}$$

Da Unter- und Obersummen gegen denselben Wert konvergieren, existiert das Integral und ist gleich diesem Wert.

```
hold(int(x^2,x=0..b))=int(f(x), x=0..b)
```

$$\int_0^b x^2 \, dx = \frac{b^3}{3}$$

Auch die Summen kann man symbolisch erhalten:

```
student::riemann(f(x),x=0..4,4, Left );float(%);
student::riemann(f(x),x=0..4,4 );float(%);
student::riemann(f(x),x=0..4,4, Right );float(%);
```

$$\sum_{i4=0}^{3} i4^2$$

$$14.0$$

$$\sum_{i5=0}^{3} \left(i5 + \frac{1}{2}\right)^2$$

$$21.0$$

$$\sum_{i6=1}^{4} i6^2$$

$$30.0$$

```
float(student::riemann(f(x),x=0..4,4*n )) $ n=1..5;
```

21.0, 21.25, 21.2962963, 21.3125, 21.32

```
float(int(f(x), x=0..4))
```

21.33333333

##

bestimmtes Intergal über x^2 von a bis b

```
hold(int(x^2,x=a..b))=int(x^2, x=a..b)
```

$$\int_a^b x^2 \, dx = \frac{b^3}{3} - \frac{a^3}{3}$$

Das Entsprechende für andere Funktionen:

```
hold(int(1,x=a..b))=int(1, x=a..b);
hold(int(x,x=a..b))=int(x, x=a..b);
hold(int(x^2,x=a..b))=int(x^2, x=a..b);
```

```
hold(int(x^3,x=a..b))=int(x^3, x=a..b);
hold(int(x^4,x=a..b))=int(x^4, x=a..b);
hold(int(sin(x),x=a..b))=int(sin(x), x=a..b);
hold(int(cos(x),x=a..b))=int(cos(x), x=a..b);
```

$$\int_a^b 1\,dx = b - a$$

$$\int_a^b x\,dx = \frac{b^2}{2} - \frac{a^2}{2}$$

$$\int_a^b x^2\,dx = \frac{b^3}{3} - \frac{a^3}{3}$$

$$\int_a^b x^3\,dx = \frac{b^4}{4} - \frac{a^4}{4}$$

$$\int_a^b x^4\,dx = \frac{b^5}{5} - \frac{a^5}{5}$$

$$\int_a^b \sin(x)\,dx = \cos(a) - \cos(b)$$

$$\int_a^b \cos(x)\,dx = \sin(b) - \sin(a)$$